
*---------------------------------; 

*            EXAMPLE #1           ; 

*---------------------------------; 

* Has Both - FORMAT and INFORMAT. ; 

*---------------------------------; 

*  NOTE: Writing and Reading a    ; 

*        "Good" DATE.             ; 

*---------------------------------; 

data _null_ ; 

   MSUGday = today() ; 

   put MSUGday = date9. ; 

run ; 

 

data _null_ ; 

   input MSUGday : date9. ; 

   put MSUGday = WORDDATE. ; 

   cards ; 

05Nov2009 

; 

run ; 

*---------------------------------; 

*            EXAMPLE #2           ; 

*---------------------------------; 

* Has a FORMAT but NO INFORMAT.   ; 

*---------------------------------; 

*  NOTE: Reading a "Not So Good"  ; 

*        DATE.                    ; 

*---------------------------------; 

 

data _null_ ; 

   input @1 txtdate $char20. ; 

   part1 = scan(txtdate,2,', ') ; 

   part2 = substr(txtdate,1,3) ; 

   part3 = scan(txtdate,3,', ') ; 

   parts = trim(part1)|| 

           trim(part2)|| 

           trim(part3) ; 

   put part1= part2= part3= parts= ; 

   MSUGday = input (parts,date9.) ; 

   put MSUGday = MMDDYY10. ; 

   cards ; 

November 5, 2009 

; 

run ; 

*---------------------------------; 

*            EXAMPLE #3           ; 

*---------------------------------; 

* Has a FORMAT but NO INFORMAT.   ; 

*---------------------------------; 

*  NOTE: Reading a "Not So Good"  ; 

*        DATE with Your Own       ; 

*        INFORMAT.                ; 

*---------------------------------; 

 

data cntlin ; 

   retain fmtname "baddate" type "I" ; 

   do label = "01Jan2009"d to  

              "31dec2009"d ; 

      start = trim(put(label,worddate.-L)) ; 

      output ; 

   end ; 

run ; 

 

proc format cntlin = cntlin ; 

run ; 

 

data _null_ ; 

   input d baddate. ; 

   put d = : mmddyy10. ; 

   cards ; 

November 5, 2009 

; 

run ; 

*---------------------------------; 

*            EXAMPLE #4           ; 

*---------------------------------; 

* Has NO FORMAT and NO INFORMAT.  ; 

*---------------------------------; 

*  NOTE: Reading a "Strange"      ; 

*        DATE with Your Own       ; 

*        INFORMAT.                ; 

*---------------------------------; 

proc format ; 

   picture STRANGE low-high = 

   '%Y/%b/%0d' 

   (datatype = date) ; 

run ; 

 

data cntlin2 ; 

   retain fmtname "MyNewDt" type "I" ; 

   do label = "01Jan2009"d to  

              "31dec2009"d ; 

      start = put(label,STRANGE11.-L) ; 

      output ; 

   end ; 

   format label date9. ; 

run ; 

 

proc format cntlin = cntlin2 ; 

run ; 

 

data _null_ ; 

   input MSUGday MyNewDt. ; 

   put MSUGday = : worddate. ; 

   cards ; 

2009/NOV/05 

run ; 

*---------------------------------; 

*            EXAMPLE #5           ; 

*---------------------------------; 

* Has FORMAT and INFORMAT but any ; 

* of 9 INFORMAT values occur.     ; 

*---------------------------------; 

*  NOTE: Reading a "Mixed Up"     ; 

*        DATE with ANYDTDTE       ; 

*        INFORMAT.                ; 

*        Notice the ambiguity in  ; 

*        the 3rd and the 5th      ; 

*        values                   ;  

*---------------------------------; 

 

*options datestyle = MDY ; 

 

data _null_ ; 

  input checkdate : anydtdte. ; 

  put checkdate=:date9. ; 

  cards ; 

05NOV2009 

05NOV2009 14:30:12.9 

05112009 

20091105 

11052009 

NOV2009 

14:30:12.9 

20091105 

05Q4 

run ; 

*---------------------------------; 

*            EXAMPLE #6           ; 

*---------------------------------; 

* Invalid DATE values do not      ; 

* receive a INVALID NOTE with     ; 

* ANYDTDTE but it does with       ; 

* other date INFORMATS.           ; 

*---------------------------------; 

*  NOTE: DATE. Informat used.     ; 

*---------------------------------; 

 

/* Example with ANYDTDTE */ 

data _null_ ; 

  input mydate anydtdte20. ; 

  put mydate=:date9. ; 

  cards ; 

05November2009 

November 5, 2009 

run ; 

 

/* Example with DATE */ 

data _null_ ; 

  input mydate : DATE14. ; 

  put mydate=:date9. ; 

  cards ; 

05November2009 

run ; 

*---------------------------------; 

*            EXAMPLE #7           ; 

*---------------------------------; 

* ANYDTDTE also reads impossible  ; 

* date values and reports a wrong ; 

* date.                           ; 

*---------------------------------; 

*  NOTE: 9.2 has NOT corrected.   ; 

*---------------------------------; 

 

/* Bugs */ 

data _null_ ; 

  dob=input("110009",anydtdte6.); 

  put dob mmddyy10.; 

run; 

*---------------------------------; 

*            EXAMPLE #8           ; 

*---------------------------------; 

* ANYDTDTE has trouble reading    ; 

* values after the 15th and before; 

* the 22nd when DDMMYYYY INFORMAT ; 

*---------------------------------; 

*  NOTE: 9.2 has corrected.       ; 

*---------------------------------; 

 

data _null_ ; 

  input date anydtdte8.; 

  format date mmddyy10.; 

  put date= ; 

  datalines; 

11052009 

15112009 

16112009 

17112009 

21112009 

22112009 

run ; 

*---------------------------------; 

*            EXAMPLE #9           ; 

*---------------------------------; 

* The WORDDATE value is not a     ; 

* documented INFORMAT but works   ; 

* with ANYDTDTE.                  ; 

*---------------------------------; 

*  NOTE: Unintended "Feature" and ; 

*        a good one.              ; 

*---------------------------------; 

 

data _null_ ; 

  input mydate anydtdte20.; 

  format mydate date9.; 

  put mydate = ; 

  cards ; 

November 5, 2009 

run; 


